

Soluções de EDOs em Séries de Potências

FERNANDO DEEKE SASSE
Departamento de Matemática
CCT - UDESC

Equação de Legendre

Consideremos a equação de Legendre

$$(1 - x^2) \frac{d^2}{dx^2} y(x) - 2x \frac{d}{dx} y(x) + l(l + 1) y(x) = 0.$$

Busquemos uma solução em série de potências em torno do ponto ordinário $x = 0$.

```
> restart;
```

```
> y := sum(a[i]*x^i, i = k-2 .. k+2);
```

$$y := a_{k-2} x^{k-2} + a_{-1+k} x^{-1+k} + a_k x^k + a_{1+k} x^{1+k} + a_{k+2} x^{k+2} \quad (1.1)$$

```
> eq1 := (1-x^2)*(diff(y, x, x)) - 2*x*(diff(y, x)) + l*(l+1)*y = 0;
```

```
> eq2 := simplify(eq1);
```

$$\begin{aligned} eq2 := & -a_k x^k k^2 - 2 a_{k-2} x^{k-2} - 2 a_{1+k} x^{1+k} - 6 a_{k+2} x^{k+2} + a_k x^{k-2} k^2 \\ & - a_k x^{k-2} k - x^{k-2} a_{k-2} k^2 - x^{-1+k} a_{-1+k} k^2 - 3 x^{1+k} a_{1+k} k \\ & - x^{1+k} a_{1+k} k^2 - x^{k+2} a_{k+2} k^2 + x^{-1+k} a_{-1+k} k - 5 x^{k+2} a_{k+2} k \\ & + 3 x^{k-2} a_{k-2} k + a_{k-2} x^{k-4} k^2 - 5 a_{k-2} x^{k-4} k - 3 a_{-1+k} x^{-3+k} k \\ & + a_{-1+k} x^{-3+k} k^2 + a_{1+k} x^{-1+k} k + a_{1+k} x^{-1+k} k^2 + a_{k+2} x^k k^2 \\ & + 3 a_{k+2} x^k k + l a_{k-2} x^{k-2} + l a_{-1+k} x^{-1+k} + l a_k x^k + l a_{1+k} x^{1+k} \\ & + l a_{k+2} x^{k+2} + l^2 a_{k-2} x^{k-2} + l^2 a_{-1+k} x^{-1+k} + l^2 a_k x^k + l^2 a_{1+k} x^{1+k} \\ & + l^2 a_{k+2} x^{k+2} - a_k x^k k + 6 a_{k-2} x^{k-4} + 2 a_{-1+k} x^{-3+k} + 2 a_{k+2} x^k = 0 \end{aligned} \quad (1.2)$$

```
> eq3 := map(coeff, eq2, x^k);
```

$$eq3 := 2 a_{k+2} - a_k k^2 + l a_k + 3 a_{k+2} k - a_k k + l^2 a_k + a_{k+2} k^2 = 0 \quad (1.3)$$

A relação de recorrência é dada por

```
> eq4 := isolate(eq3, a[k+2]);
```

$$eq4 := a_{k+2} = \frac{a_k k^2 - l a_k + a_k k - l^2 a_k}{2 + 3 k + k^2} \quad (1.4)$$

Quando l é inteiro uma das séries sempre termina. Calculemos explicitamente as séries neste caso.

```
> restart;
```

```
> y := sum(a[i]*x^i, i = 0 .. 8);
```

$$(1.5)$$

$$y := a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6 + a_7 x^7 + a_8 x^8 \quad (1.5)$$

$$\begin{aligned} &> \text{eq1} := (1-x^2) * (\text{diff}(y, x, x)) - 2*x*(\text{diff}(y, x)) + 1*(l+1)*y = 0; \\ \text{eq1} &:= (1-x^2) (2 a_2 + 6 a_3 x + 12 a_4 x^2 + 20 a_5 x^3 + 30 a_6 x^4 + 42 a_7 x^5 + 56 a_8 x^6) \end{aligned} \quad (1.6)$$

$$\begin{aligned} &- 2 x (a_1 + 2 a_2 x + 3 a_3 x^2 + 4 a_4 x^3 + 5 a_5 x^4 + 6 a_6 x^5 + 7 a_7 x^6 + 8 a_8 x^7) \\ &+ l(l+1) (a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6 + a_7 x^7 + a_8 x^8) = 0 \end{aligned}$$

$$\begin{aligned} &> \text{eq2} := \text{collect}(\text{eq1}, x); \\ \text{eq2} &:= (-72 a_8 + l(l+1) a_8) x^8 + (-56 a_7 + l(l+1) a_7) x^7 + (56 a_8 - 42 a_6 + l(l \end{aligned} \quad (1.7)$$

$$\begin{aligned} &+ 1) a_6) x^6 + (-30 a_5 + 42 a_7 + l(l+1) a_5) x^5 + (l(l+1) a_4 - 20 a_4 \\ &+ 30 a_6) x^4 + (-12 a_3 + 20 a_5 + l(l+1) a_3) x^3 + (12 a_4 - 6 a_2 + l(l \\ &+ 1) a_2) x^2 + (6 a_3 + l(l+1) a_1 - 2 a_1) x + 2 a_2 + l(l+1) a_0 = 0 \end{aligned}$$

```
> for i from 0 to 6 do
>   g:=map(coeff,eq2,x,i);
   isolate(g,a[i+2]);
   assign(%);
od;
```

$$g := 2 a_2 + l(l+1) a_0 = 0$$

$$a_2 = -\frac{1}{2} l(l+1) a_0$$

$$g := 6 a_3 + l(l+1) a_1 - 2 a_1 = 0$$

$$a_3 = -\frac{1}{6} l(l+1) a_1 + \frac{1}{3} a_1$$

$$g := 12 a_4 + 3 l(l+1) a_0 - \frac{1}{2} l^2 (l+1)^2 a_0 = 0$$

$$a_4 = -\frac{1}{4} l(l+1) a_0 + \frac{1}{24} l^2 (l+1)^2 a_0$$

$$g := 2 l(l+1) a_1 - 4 a_1 + 20 a_5 + l(l+1) \left(-\frac{1}{6} l(l+1) a_1 + \frac{1}{3} a_1 \right) = 0$$

$$a_5 = -\frac{1}{10} l(l+1) a_1 + \frac{1}{5} a_1 - \frac{1}{20} l(l+1) \left(-\frac{1}{6} l(l+1) a_1 + \frac{1}{3} a_1 \right)$$

$$\begin{aligned} g &:= l(l+1) \left(-\frac{1}{4} l(l+1) a_0 + \frac{1}{24} l^2 (l+1)^2 a_0 \right) + 5 l(l+1) a_0 - \frac{5}{6} l^2 (l \\ &+ 1)^2 a_0 + 30 a_6 = 0 \end{aligned}$$

$$a_6 = -\frac{1}{30} l(l+1) \left(-\frac{1}{4} l(l+1) a_0 + \frac{1}{24} l^2 (l+1)^2 a_0 \right) - \frac{1}{6} l(l+1) a_0$$

$$+ \frac{1}{36} l^2 (l+1)^2 a_0$$

$$\begin{aligned}
g &:= 3l(l+1)a_1 - 6a_1 + \frac{3}{2}l(l+1)\left(-\frac{1}{6}l(l+1)a_1 + \frac{1}{3}a_1\right) + 42a_7 + l(l \\
&\quad + 1)\left(-\frac{1}{10}l(l+1)a_1 + \frac{1}{5}a_1 - \frac{1}{20}l(l+1)\left(-\frac{1}{6}l(l+1)a_1 + \frac{1}{3}a_1\right)\right) = 0 \\
a_7 &= -\frac{1}{14}l(l+1)a_1 + \frac{1}{7}a_1 - \frac{1}{28}l(l+1)\left(-\frac{1}{6}l(l+1)a_1 + \frac{1}{3}a_1\right) - \frac{1}{42}l(l \\
&\quad + 1)\left(-\frac{1}{10}l(l+1)a_1 + \frac{1}{5}a_1 - \frac{1}{20}l(l+1)\left(-\frac{1}{6}l(l+1)a_1 + \frac{1}{3}a_1\right)\right) \\
g &:= 56a_8 + \frac{7}{5}l(l+1)\left(-\frac{1}{4}l(l+1)a_0 + \frac{1}{24}l^2(l+1)^2a_0\right) + 7l(l+1)a_0 \\
&\quad - \frac{7}{6}l^2(l+1)^2a_0 + l(l+1)\left(-\frac{1}{30}l(l+1)\left(-\frac{1}{4}l(l+1)a_0 + \frac{1}{24}l^2(l \\
&\quad + 1)^2a_0\right) - \frac{1}{6}l(l+1)a_0 + \frac{1}{36}l^2(l+1)^2a_0\right) = 0 \\
a_8 &= -\frac{1}{40}l(l+1)\left(-\frac{1}{4}l(l+1)a_0 + \frac{1}{24}l^2(l+1)^2a_0\right) - \frac{1}{8}l(l+1)a_0 \tag{1.8} \\
&\quad + \frac{1}{48}l^2(l+1)^2a_0 - \frac{1}{56}l(l+1)\left(-\frac{1}{30}l(l+1)\left(-\frac{1}{4}l(l+1)a_0 \right. \right. \\
&\quad \left. \left. + \frac{1}{24}l^2(l+1)^2a_0\right) - \frac{1}{6}l(l+1)a_0 + \frac{1}{36}l^2(l+1)^2a_0\right)
\end{aligned}$$

Para l inteiro a série termina. Para cada valor de l temos um polinômio, denominado polinômio de Legendre. Para l par temos

```

> a[1] := 0; a[0] := 1;
> for l from 0 by 2 to 8 do P[l] := y end do;

```

$$\begin{aligned}
P_0 &:= 1 \\
P_2 &:= 1 - 3x^2 \\
P_4 &:= 1 - 10x^2 + \frac{35}{3}x^4 \\
P_6 &:= 1 - 21x^2 + 63x^4 - \frac{231}{5}x^6 \\
P_8 &:= 1 - 36x^2 + 198x^4 - \frac{1716}{5}x^6 + \frac{1287}{7}x^8 \tag{1.9}
\end{aligned}$$

```

> a[1] := 1; a[0] := 0;

```

$$\begin{aligned}
a_1 &:= 1 \\
a_0 &:= 0 \tag{1.10}
\end{aligned}$$

Para l ímpar,

```

> for l from 1 by 2 to 7 do P[l] := y end do;

```


$$\begin{aligned}
P_1 &:= x \\
P_3 &:= x - \frac{5}{3}x^3
\end{aligned}$$

$$P_5 := x - \frac{14}{3}x^3 + \frac{21}{5}x^5$$

$$P_7 := x - 9x^3 + \frac{99}{5}x^5 - \frac{429}{35}x^7 \quad (1.11)$$

Façamos os respectivos gráficos:

```
> plot([seq(P[j], j=0..8)], x=-1..1);
```



No Maple os polinômios de Legendre podem ser gerados da seguinte forma:

```
> for j from 0 to 8 do
  p[j] := expand(LegendreP(j, x));
od;
```

$$p_0 := 1$$

$$p_1 := x$$

$$p_2 := -\frac{1}{2} + \frac{3}{2}x^2$$

$$p_3 := \frac{5}{2}x^3 - \frac{3}{2}x$$

$$\begin{aligned}
p_4 &:= \frac{3}{8} + \frac{35}{8}x^4 - \frac{15}{4}x^2 \\
p_5 &:= \frac{63}{8}x^5 - \frac{35}{4}x^3 + \frac{15}{8}x \\
p_6 &:= -\frac{5}{16} + \frac{231}{16}x^6 - \frac{315}{16}x^4 + \frac{105}{16}x^2 \\
p_7 &:= \frac{429}{16}x^7 - \frac{693}{16}x^5 + \frac{315}{16}x^3 - \frac{35}{16}x \\
p_8 &:= \frac{35}{128} + \frac{6435}{128}x^8 - \frac{3003}{32}x^6 + \frac{3465}{64}x^4 - \frac{315}{32}x^2
\end{aligned} \tag{1.12}$$

Notemos que estes polinômios diferem daqueles calculados anteriormente somente por fatores constantes.

Vejamos como o Maple resolve a equação de Legendre com $l = 8$:

```
> x:='x':
> l := 8:
> Eq1 := (1-x^2)*(diff(Y(x), x, x))-2*x*(diff(Y(x), x))+l*(l+1)*Y(x) = 0;
```

$$Eq1 := (1 - x^2) \left(\frac{d^2}{dx^2} Y(x) \right) - 2x \left(\frac{d}{dx} Y(x) \right) + 72 Y(x) = 0 \tag{1.13}$$

```
> dsolve(Eq1, Y(x));
```

$$\begin{aligned}
Y(x) = & _C1 \left(\frac{35}{128} + \frac{6435}{128}x^8 - \frac{3003}{32}x^6 + \frac{3465}{64}x^4 - \frac{315}{32}x^2 \right) \\
& + _C2 \left(\frac{1}{8960} (1225 + 225225x^8 - 420420x^6 + 242550x^4 \right. \\
& \left. - 44100x^2) \ln\left(\frac{-x-1}{x-1}\right) - \frac{4213}{128}x^3 + \frac{15159}{4480}x - \frac{6435}{128}x^7 + \frac{9867}{128}x^5 \right)
\end{aligned} \tag{1.14}$$

```
> yy := op(2, %);
```

$$\begin{aligned}
yy := & _C1 \left(\frac{35}{128} + \frac{6435}{128}x^8 - \frac{3003}{32}x^6 + \frac{3465}{64}x^4 - \frac{315}{32}x^2 \right) + _C2 \left(\frac{1}{8960} (1225 \right. \\
& \left. + 225225x^8 - 420420x^6 + 242550x^4 - 44100x^2) \ln\left(\frac{-x-1}{x-1}\right) - \frac{4213}{128}x^3 \right. \\
& \left. + \frac{15159}{4480}x - \frac{6435}{128}x^7 + \frac{9867}{128}x^5 \right)
\end{aligned} \tag{1.15}$$

Apliquemos condições iniciais e façamos o correspondente gráfico.

```
> EE1:=subs(x=0.8,yy)=1;
EE1 := -0.016655300 _C1 + _C2 (-0.008327651786 ln(9.000000000) + 0.57138029) = 1 \tag{1.16}
```


```
> EE2 := subs(x = 0.8, diff(yy, x)) = -2;
EE2 := -5.02948800 _C1 + _C2 (-2.514743996 ln(9.000000000) + 5.7622806) = -2 \tag{1.17}
```

```
> sol := solve({EE1, EE2});
sol := {_C1 = 0.4834758325, _C2 = 1.822607495} \tag{1.18}
```

```

> assign(sol);
> yy
0.1322004229 + 24.30599205 x8 - 45.37118516 x6 + 26.17568374 x4 - 4.759215226 x2 (1.19)
+ 0.0002034160151 (1225 + 225225 x8 - 420420 x6 + 242550 x4
- 44100 x2) ln( (-x-1)/(x-1) ) - 59.98941700 x3 + 6.167166745 x - 91.62874399 x7
+ 140.4974074 x5
> plot(yy, x = -0.99 .. 0.99);

```



Notemos que esta solução pode ser gerada através de "força bruta", usando a relação de recorrência obtida anteriormente:

```

> 1;

```

8 (1.20)

```

> ak+2 =

```

$$a_{k+2} = \frac{a_k k^2 - 1 a_k + a_k k - 1^2 a_k}{2 + 3 k + k^2}$$

$$a_{k+2} = \frac{a_k k - 72 a_k + a_k k^2}{2 + 3 k + k^2} \quad (1.21)$$

> $k := 'k'$; $a_1 := 'a_1'$; $a_0 := 'a_0'$;

> **for** k **to** 20 **do**

$$a_{k+2} := \frac{a_k k^2 - l a_k + a_k k - l^2 a_k}{2 + 3k + k^2}$$

od;

$$a_3 := -\frac{35}{3} a_1$$

$$a_4 := 198 a_0$$

$$a_5 := 35 a_1$$

$$a_6 := -\frac{1716}{5} a_0$$

$$a_7 := -35 a_1$$

$$a_8 := \frac{1287}{7} a_0$$

$$a_9 := \frac{70}{9} a_1$$

$$a_{10} := 0$$

$$a_{11} := \frac{14}{11} a_1$$

$$a_{12} := 0$$

$$a_{13} := \frac{70}{143} a_1$$

$$a_{14} := 0$$

$$a_{15} := \frac{10}{39} a_1$$

$$a_{16} := 0$$

$$a_{17} := \frac{35}{221} a_1$$

$$a_{18} := 0$$

$$a_{19} := \frac{35}{323} a_1$$

$$a_{20} := 0$$

$$a_{21} := \frac{77}{969} a_1$$

$$a_{22} := 0$$

(1.22)

Vamos gerar coeficientes da série até $O(20)$:

A série é então dada por

```
> sum(a[m]*x^m,m=0..20);
```

$$a_0 + a_1 x - 36 a_0 x^2 - \frac{35}{3} a_1 x^3 + 198 a_0 x^4 + 35 a_1 x^5 - \frac{1716}{5} a_0 x^6 - 35 a_1 x^7 \quad (1.23)$$
$$+ \frac{1287}{7} a_0 x^8 + \frac{70}{9} a_1 x^9 + \frac{14}{11} a_1 x^{11} + \frac{70}{143} a_1 x^{13} + \frac{10}{39} a_1 x^{15} + \frac{35}{221} a_1 x^{17}$$
$$+ \frac{35}{323} a_1 x^{19}$$

```
> collect(%,a[1]):Y:=collect(%,a[0]);
```

$$Y := \left(1 - 36 x^2 + 198 x^4 - \frac{1716}{5} x^6 + \frac{1287}{7} x^8 \right) a_0 + \left(x - \frac{35}{3} x^3 + 35 x^5 - 35 x^7 \right. \quad (1.24)$$
$$\left. + \frac{70}{9} x^9 + \frac{14}{11} x^{11} + \frac{70}{143} x^{13} + \frac{10}{39} x^{15} + \frac{35}{221} x^{17} + \frac{35}{323} x^{19} \right) a_1$$

Apliquemos agora as mesmas condições iniciais e façamos o gráfico, comparando com a solução dada pelo Maple:

```
> ee1:=subs(x=0.8,Y)=1;
```

$$ee1 := -0.06091081 a_0 + 0.1497373445 a_1 = 1 \quad (1.25)$$

```
> ee2 := subs(x = 0.8, diff(Y, x)) = -2;
```


$$ee2 := -18.3935561 a_0 + 0.02134134212 a_1 = -2 \quad (1.26)$$

```
> sol := solve({ee1, ee2});
```

$$sol := \{a_0 = 0.1165373823, a_1 = 6.725766306\} \quad (1.27)$$

```
> assign(sol);
```

```
> plot([yy, Y], x = -0.9 .. 0.9, color = [blue, red], legend =  
["Solução exata gerada pelo Maple", "Série gerada por  
recorrência"]);
```



▼ Método de Frobenius

Consideremos a seguinte EDO linear:

$$P(x) \frac{d^2}{dx^2} y(x) + Q(x) \frac{d}{dx} y(x) + R(x)y(x) = 0.$$

Suponhamos que $x = 0$ é um ponto singular regular desta equação, ou seja, $P(0) = 0$, e $x \frac{Q(x)}{P(x)}$, $x^2 \frac{R(x)}{P(x)}$ são analíticos em $x = 0$.

Buscaremos uma solução desta equação da forma

$$y(x) = \sum_{n=0}^{\infty} c_n x^{n+r},$$

de modo que

$$\frac{d}{dx} y(x) = \sum_{n=0}^{\infty} (n+r) c_n x^{n+r-1},$$

$$\frac{d^2}{dx^2} y(x) = \sum_{n=0}^{\infty} (n+r)(n+r-1) c_n x^{n+r-2}.$$

Substituindo estas séries na EDO temos

$$P(x) \left(\sum_{n=0}^{\infty} (n+r)(n+r-1) c_n x^{n+r-2} \right) + Q(x) \left(\sum_{n=0}^{\infty} (n+r) c_n x^{n+r-1} \right) + \sum_{n=0}^{\infty} c_n x^{n+r} = 0.$$

Definimos

$$S_2 = P(x) \left(\sum_{n=0}^{\infty} (n+r)(n+r-1) c_n x^{n+r-2} \right),$$

$$S_1 = Q(x) \left(\sum_{n=0}^{\infty} (n+r) c_n x^{n+r-1} \right),$$

$$S_0 = \sum_{n=0}^{\infty} c_n x^{n+r},$$

de modo que a EDO torna-se $S_2 + S_1 + S_0 = 0$. Consideremos alguns exemplos específicos.

Exemplo 1

Busquemos a solução da equação

$$3x \frac{d^2}{dx^2} y(x) + \frac{d}{dx} y(x) - y(x) = 0.$$

em torno do ponto singular regular $x = 0$.

```
> restart;
> P := 3*x: Q:=1: R:=-1:
> S2 := sum(simplify(P*c[n]*(n+r)*(n+r-1)*x^(n+r-2)), n = 0 .. k)
;
```

$$S2 := \sum_{n=0}^k 3 x^{n+r-1} c_n (n+r) (n+r-1) \quad (3.1)$$

```
> S1 := sum(simplify(Q*c[n]*(n+r)*x^(n+r-1)), n = 0 .. k);
```

$$S1 := \sum_{n=0}^k x^{n+r-1} c_n (n+r) \quad (3.2)$$

```
> S0 := sum(simplify(R*c[n]*x^(n+r)), n = 0 .. k);
```

$$S0 := \sum_{n=0}^k (-c_n x^{n+r}) \quad (3.3)$$

A equação pode agora ser escrita como

```
> S := S2+S1+S0;
```

$$S := \sum_{n=0}^k 3 x^{n+r-1} c_n (n+r) (n+r-1) + \sum_{n=0}^k x^{n+r-1} c_n (n+r) + \sum_{n=0}^k (-c_n x^{n+r}) \quad (3.4)$$

```
> k := 6;
```

$$k := 6 \quad (3.5)$$

```
> eq1 := simplify(S/x^r);
```

$$\begin{aligned} eq1 := & \frac{1}{x} (96 c_6 x^6 + c_1 x + 21 c_3 x^3 + 8 c_2 x^2 + 40 c_4 x^4 + 65 c_5 x^5 - 2 c_0 r + 3 c_0 r^2 \\ & + 10 x^2 c_2 r + 3 x^2 c_2 r^2 + 16 x^3 c_3 r + 3 x^3 c_3 r^2 + 22 x^4 c_4 r + 3 x^4 c_4 r^2 \\ & + 28 x^5 c_5 r + 3 x^5 c_5 r^2 + 34 x^6 c_6 r + 3 x^6 c_6 r^2 + 3 c_1 r^2 x + 4 c_1 r x - c_0 x \\ & - c_6 x^7 - c_1 x^2 - c_3 x^4 - c_2 x^3 - c_4 x^5 - c_5 x^6) \end{aligned} \quad (3.6)$$

```
> eq2 := x*%:
```

```
> eq3 := collect(%, x);
```

$$\begin{aligned} eq3 := & -c_6 x^7 + (96 c_6 + 3 c_6 r^2 + 34 c_6 r - c_5) x^6 + (-c_4 + 28 c_5 r + 65 c_5 \\ & + 3 c_5 r^2) x^5 + (3 c_4 r^2 + 40 c_4 - c_3 + 22 c_4 r) x^4 + (-c_2 + 21 c_3 + 16 c_3 r \\ & + 3 c_3 r^2) x^3 + (10 c_2 r - c_1 + 3 c_2 r^2 + 8 c_2) x^2 + (-c_0 + c_1 + 3 c_1 r^2 + 4 c_1 r) x \\ & - 2 c_0 r + 3 c_0 r^2 \end{aligned} \quad (3.7)$$

```
> e0 := coeff(eq3, x, 0);
```

$$e0 := -2 c_0 r + 3 c_0 r^2 \quad (3.8)$$

```
> s := solve(e0, r);
```

$$(3.9)$$

$$s := 0, \frac{2}{3} \quad (3.9)$$

```
> r1 := s[1]; r2 := s[2];
```

$$r1 := 0$$

$$r2 := \frac{2}{3} \quad (3.10)$$

Como $r2 - r1 = 2/3$, as relações de recorrências associadas a cada raiz geram soluções linearmente independentes. Determinemos as relações de recorrência para r indeterminado:

```
> eqs := array(1 .. 6);
for n from 1 to 6 do
    eqs[n] := coeff(eq3, x, n) = 0
end do;
coeffEqs := convert(eqs, list);
```

```
eqs := array(1..6, [ ])
```

$$eqs_1 := -c_0 + c_1 + 3 c_1 r^2 + 4 c_1 r = 0$$

$$eqs_2 := 10 c_2 r - c_1 + 3 c_2 r^2 + 8 c_2 = 0$$

$$eqs_3 := -c_2 + 21 c_3 + 16 c_3 r + 3 c_3 r^2 = 0$$

$$eqs_4 := 3 c_4 r^2 + 40 c_4 - c_3 + 22 c_4 r = 0$$

$$eqs_5 := -c_4 + 28 c_5 r + 65 c_5 + 3 c_5 r^2 = 0$$

$$eqs_6 := 96 c_6 + 3 c_6 r^2 + 34 c_6 r - c_5 = 0$$

$$\text{coeffEqs} := \left[-c_0 + c_1 + 3 c_1 r^2 + 4 c_1 r = 0, 10 c_2 r - c_1 + 3 c_2 r^2 + 8 c_2 = 0, -c_2 + 21 c_3 \right. \quad (3.11) \\ \left. + 16 c_3 r + 3 c_3 r^2 = 0, 3 c_4 r^2 + 40 c_4 - c_3 + 22 c_4 r = 0, -c_4 + 28 c_5 r + 65 c_5 \right. \\ \left. + 3 c_5 r^2 = 0, 96 c_6 + 3 c_6 r^2 + 34 c_6 r - c_5 = 0 \right]$$

Temos aqui 6 equações para 6 incógnitas c_1 a c_6 .

```
> s1 := [seq(c[n], n = 1 .. 6)];
```

$$s1 := [c_1, c_2, c_3, c_4, c_5, c_6] \quad (3.12)$$

```
>
```

Resolvendo estas equações para a_0 obtemos

```
> coef := solve(coeffEqs, s1);
```

$$\text{coef} := \left[\left[c_1 = \frac{c_0}{1 + 3 r^2 + 4 r}, c_2 = \frac{c_0}{42 r + 67 r^2 + 8 + 42 r^3 + 9 r^4}, c_3 \right. \right. \quad (3.13) \\ \left. \left. = \frac{c_0}{1010 r + 2103 r^2 + 2080 r^3 + 168 + 270 r^5 + 27 r^6 + 1062 r^4}, c_4 = c_0 / \right. \right]$$

$$\begin{aligned}
& (81 r^8 + 1404 r^7 + 10206 r^6 + 40404 r^5 + 94549 r^4 + 132496 r^3 + 106844 r^2 \\
& + 44096 r + 6720), c_5 = c_0 / (243 r^{10} + 6480 r^9 + 75195 r^8 + 498240 r^7 \\
& + 2078349 r^6 + 5671120 r^5 + 10176105 r^4 + 11736160 r^3 + 8199708 r^2 \\
& + 3054400 r + 436800), c_6 = c_0 / (41932800 + 892331968 r^2 + 308073600 r \\
& + 925623570 r^5 + 1400534644 r^4 + 422867899 r^6 + 135508266 r^7 + 30393927 r^8 \\
& + 4673430 r^9 + 469233 r^{10} + 27702 r^{11} + 729 r^{12} + 1414624632 r^3)]]
\end{aligned}$$

> assign(coef)

> **y := x^r*(sum(c[i]*x^i, i = 0 .. 6));**

$$y := x^r \left(c_0 + \frac{c_0 x}{1 + 3 r^2 + 4 r} + \frac{c_0 x^2}{42 r + 67 r^2 + 8 + 42 r^3 + 9 r^4} \right.$$

$$+ \frac{c_0 x^3}{1010 r + 2103 r^2 + 2080 r^3 + 168 + 270 r^5 + 27 r^6 + 1062 r^4} + (c_0 x^4) /$$

$$\begin{aligned}
& (81 r^8 + 1404 r^7 + 10206 r^6 + 40404 r^5 + 94549 r^4 + 132496 r^3 + 106844 r^2 \\
& + 44096 r + 6720) + (c_0 x^5) / (243 r^{10} + 6480 r^9 + 75195 r^8 + 498240 r^7 \\
& + 2078349 r^6 + 5671120 r^5 + 10176105 r^4 + 11736160 r^3 + 8199708 r^2 \\
& + 3054400 r + 436800) + (c_0 x^6) / (41932800 + 892331968 r^2 + 308073600 r \\
& + 925623570 r^5 + 1400534644 r^4 + 422867899 r^6 + 135508266 r^7 + 30393927 r^8 \\
& + 4673430 r^9 + 469233 r^{10} + 27702 r^{11} + 729 r^{12} + 1414624632 r^3))
\end{aligned}$$

> y1 := simplify((subs(r=r1, y)) / c[0])

$$y1 := 1 + x + \frac{1}{8} x^2 + \frac{1}{168} x^3 + \frac{1}{6720} x^4 + \frac{1}{436800} x^5 + \frac{1}{41932800} x^6$$

> y2 := simplify((subs(r=r2, y)) / c[0])

$$y2 := \frac{1}{1507968000} x^{2/3} (1507968000 + 301593600 x + 18849600 x^2 + 571200 x^3$$

$$+ 10200 x^4 + 120 x^5 + x^6)$$

> Y := C1·y1 + C2·y2

$$\begin{aligned}
 Y := & CI \left(1 + x + \frac{1}{8} x^2 + \frac{1}{168} x^3 + \frac{1}{6720} x^4 + \frac{1}{436800} x^5 + \frac{1}{41932800} x^6 \right) \\
 & + \frac{1}{1507968000} C2 x^{2/3} (1507968000 + 301593600 x + 18849600 x^2 + 571200 x^3 \\
 & + 10200 x^4 + 120 x^5 + x^6)
 \end{aligned}
 \tag{3.17}$$

Determinemos a solução correspondente às condições iniciais $y(1) = 1$ e $y'(1) = -2$.

```
> ee1 := subs(x=1, Y)=1;
```

$$ee1 := \frac{89363137}{41932800} CI + \frac{1828992721}{1507968000} C2 = 1$$
(3.18)


```
> ee2 := subs(x = 1, diff(Y, x)) = -2;
```

$$ee2 := \frac{8865041}{6988800} CI + \frac{234056443}{226195200} C2 = -2$$
(3.19)

```
> sol := solve({ee1, ee2});
```

$$sol := \left\{ CI = \frac{328230778043750400}{63233320672117861}, C2 = -\frac{524584036247040000}{63233320672117861} \right\}$$
(3.20)

```
> assign(sol);
> plot(Y, x = 0 .. 12);
```



Comparemos este resultado com aquele fornecido pelos comandos do Maple:

```
> eq := 3*x*(diff(yy(x), `x`(x, 2)))+diff(yy(x), x)-yy(x) = 0;
```

$$eq := 3x \left(\frac{d^2}{dx^2} yy(x) \right) + \frac{d}{dx} yy(x) - yy(x) = 0 \quad (3.21)$$

A solução geral é dada por

```
> dsolve(eq, yy(x));
```

$$yy(x) = _C1 x^{1/3} \text{BesselI} \left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \sqrt{x} \right) + _C2 x^{1/3} \text{BesselK} \left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \sqrt{x} \right) \quad (3.22)$$

Usando as condições iniciais temos

```
> dsolve({eq, yy(1) = 1, (D(yy))(1) = -2}, yy(x));
```


$$yy(x) = \frac{1}{9} \left(\left(-18 \text{BesselK} \left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \right) + 3 \sqrt{3} \text{BesselK} \left(\frac{1}{3}, \frac{2}{3} \sqrt{3} \right) \right) \sqrt{3} x^{1/3} \text{BesselI} \left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \sqrt{x} \right) \right) / \left(\text{BesselK} \left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \right) \text{BesselI} \left(-\frac{1}{3}, \frac{2}{3} \sqrt{3} \right) + \text{BesselK} \left(\frac{1}{3}, \frac{2}{3} \sqrt{3} \right) \text{BesselI} \left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \right) \right) - \frac{1}{9} \left(\sqrt{3} \left(\right. \right. \quad (3.23)$$

$$-18 \operatorname{BesselI}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) - 3 \sqrt{3} \operatorname{BesselI}\left(-\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) x^{1/3} \operatorname{BesselK}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \sqrt{x}\right) \Big/ \left(\operatorname{BesselK}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) \operatorname{BesselI}\left(-\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) + \operatorname{BesselK}\left(\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) \operatorname{BesselI}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) \right)$$

> **YY := op(2, %);**


$$YY := \frac{1}{9} \left(\left(-18 \operatorname{BesselK}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) + 3 \sqrt{3} \operatorname{BesselK}\left(\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) \right) \sqrt{3} x^{1/3} \operatorname{BesselI}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \sqrt{x}\right) \right) \Big/ \left(\operatorname{BesselK}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) \operatorname{BesselI}\left(-\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) + \operatorname{BesselK}\left(\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) \operatorname{BesselI}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) \right) - \frac{1}{9} \left(\sqrt{3} \left(-18 \operatorname{BesselI}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) - 3 \sqrt{3} \operatorname{BesselI}\left(-\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) \right) x^{1/3} \operatorname{BesselK}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3} \sqrt{x}\right) \right) \Big/ \left(\operatorname{BesselK}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) \operatorname{BesselI}\left(-\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) + \operatorname{BesselK}\left(\frac{1}{3}, \frac{2}{3} \sqrt{3}\right) \operatorname{BesselI}\left(\frac{2}{3}, \frac{2}{3} \sqrt{3}\right) \right) \quad (3.24)$$

> **plot([Y, YY], x = 0 .. 50, color = [blue, red], legend = ["Série com 7 termos", "Solução em termos de funções de Bessel"]);**



Notemos que a concordância é boa até $x \approx 20$.

```
> plot([Y, YY], x = 0 .. 20, color = [blue, red], legend =  
["Série com 7 termos", "Solução em termos de funções de  
Bessel"]);
```



Exemplo 2

Busquemos a solução da equação

$$x \frac{d^2}{dx^2} y(x) + 5 \frac{d}{dx} y(x) + xy(x) = 0.$$

em torno do ponto singular regular $x = 0$.

```

> restart;
> P := x: Q:=5: R:=x:
> S2 := sum(simplify(P*c[n]*(n+r)*(n+r-1)*x^(n+r-2)), n = 0 .. k)
;

```

$$S2 := \sum_{n=0}^k x^{n+r-1} c_n (n+r) (n+r-1) \quad (4.1)$$

```

> S1 := sum(simplify(Q*c[n]*(n+r)*x^(n+r-1)), n = 0 .. k);

```

$$SI := \sum_{n=0}^k 5 x^{n+r-1} c_n (n+r) \quad (4.2)$$

> S0 := sum(simplify(R*c[n]*x^(n+r)), n = 0 .. k);

$$S0 := \sum_{n=0}^k x^{n+1+r} c_n \quad (4.3)$$

A equação pode agora ser escrita como

> s := s2+s1+s0;

$$S := \sum_{n=0}^k x^{n+r-1} c_n (n+r) (n+r-1) + \sum_{n=0}^k 5 x^{n+r-1} c_n (n+r) + \sum_{n=0}^k x^{n+1+r} c_n \quad (4.4)$$

> k := 6;

$$k := 6 \quad (4.5)$$

> eq1 := simplify(S/x^r);

$$\begin{aligned} eq1 := & \frac{1}{x} (4 c_0 r + c_0 r^2 + 8 x^2 c_2 r + x^2 c_2 r^2 + 10 x^3 c_3 r + x^3 c_3 r^2 + 12 x^4 c_4 r \\ & + x^4 c_4 r^2 + 14 x^5 c_5 r + x^5 c_5 r^2 + 16 x^6 c_6 r + x^6 c_6 r^2 + 6 c_1 r x + c_1 r^2 x + 5 c_1 x \\ & + x^6 c_4 + x^4 c_2 + x^8 c_6 + 12 x^2 c_2 + 21 x^3 c_3 + 32 x^4 c_4 + 45 x^5 c_5 + 60 x^6 c_6 \\ & + x^2 c_0 + x^3 c_1 + x^5 c_3 + x^7 c_5) \end{aligned} \quad (4.6)$$

> eq2 := x*%:

> eq3 := collect(%, x);

$$\begin{aligned} eq3 := & x^8 c_6 + x^7 c_5 + (c_4 + 60 c_6 + 16 c_6 r + c_6 r^2) x^6 + (c_5 r^2 + 45 c_5 + 14 c_5 r \\ & + c_3) x^5 + (c_4 r^2 + c_2 + 12 c_4 r + 32 c_4) x^4 + (c_1 + 10 c_3 r + 21 c_3 + c_3 r^2) x^3 \\ & + (c_2 r^2 + 8 c_2 r + 12 c_2 + c_0) x^2 + (5 c_1 + 6 c_1 r + c_1 r^2) x + 4 c_0 r + c_0 r^2 \end{aligned} \quad (4.7)$$

> e0 := coeff(eq3, x, 0);

$$e0 := 4 c_0 r + c_0 r^2 \quad (4.8)$$

> s := solve(e0, r);

$$s := 0, -4 \quad (4.9)$$

> r1 := s[1]; r2 := s[2];

$$\begin{aligned} r1 & := 0 \\ r2 & := -4 \end{aligned} \quad (4.10)$$

Como $r1 - r2 = 4$, as relações de recorrências associadas a cada raiz não geram soluções linearmente independentes.

P

> r := r1;

> eqs := array(1 .. 6);

```

for n from 1 to 6 do
  eqs[n] := coeff(eq3, x, n) = 0
end do;
coeffEqs := convert(eqs, list);
  eqs := array(1..6, [ ])
  eqs1 := 5 c1 = 0
  eqs2 := 12 c2 + c0 = 0
  eqs3 := c1 + 21 c3 = 0
  eqs4 := c2 + 32 c4 = 0
  eqs5 := 45 c5 + c3 = 0
  eqs6 := c4 + 60 c6 = 0
coeffEqs := [5 c1 = 0, 12 c2 + c0 = 0, c1 + 21 c3 = 0, c2 + 32 c4 = 0, 45 c5 + c3 = 0, c4
+ 60 c6 = 0]

```

(4.11)

Temos aqui 6 equações para 6 incógnitas c_1 a c_6 .

```

> s1 := [seq(c[n], n = 1 .. 6)];
  s1 := [c1, c2, c3, c4, c5, c6]

```

(4.12)

Resolvendo estas equações para a_0 obtemos

```

> coef := solve(coeffEqs, s1);
  coef := [[c1 = 0, c2 = -1/12 c0, c3 = 0, c4 = 1/384 c0, c5 = 0, c6 = -1/23040 c0]]

```

(4.13)

```

> assign(coef)
> y1 := expand(x^r*(sum(c[i]*x^i, i = 0 .. 6))/c[0]);
  y1 := 1 - 1/12 x2 + 1/384 x4 - 1/23040 x6

```

(4.14)

Encontremos agora a segunda solução. Como $r1 - r2 = 4 > 0$, devemos buscar uma solução da forma:

$$y2 = k y1 \ln x + x^{r2} \sum_{k=0}^{\infty} b_k x^k$$

```

> y2 := B*y1*ln(x)+x^r2*(sum(b[j]*x^j, j = 0 .. 6));
y2 := B (1 - 1/12 x2 + 1/384 x4 - 1/23040 x6) ln(x)
+ (b0 + b1 x + b2 x2 + b3 x3 + b4 x4 + b5 x5 + b6 x6) / x4

```

(4.15)

Se $B = 0$

```

> r := r2
  r := -4

```

(4.16)

> `coeff(eq3, x, 2)`

$$\frac{4}{3} c_0 \quad (4.17)$$

O que implicaria $c_0 = 0$, de modo que devemos supor que $B \neq 0$.

> `eq4:=x*(diff(y2, x, x))+5*(diff(y2, x))+x*y2;`

$$\begin{aligned} eq4 := & x \left(B \left(-\frac{1}{6} + \frac{1}{32} x^2 - \frac{1}{768} x^4 \right) \ln(x) + \frac{2B \left(-\frac{1}{6} x + \frac{1}{96} x^3 - \frac{1}{3840} x^5 \right)}{x} \right. \\ & - \frac{B \left(1 - \frac{1}{12} x^2 + \frac{1}{384} x^4 - \frac{1}{23040} x^6 \right)}{x^2} \\ & + \frac{20 (b_0 + b_1 x + b_2 x^2 + b_3 x^3 + b_4 x^4 + b_5 x^5 + b_6 x^6)}{x^6} \\ & - \frac{8 (b_1 + 2 b_2 x + 3 b_3 x^2 + 4 b_4 x^3 + 5 b_5 x^4 + 6 b_6 x^5)}{x^5} \\ & \left. + \frac{2 b_2 + 6 b_3 x + 12 b_4 x^2 + 20 b_5 x^3 + 30 b_6 x^4}{x^4} \right) + 5 B \left(-\frac{1}{6} x + \frac{1}{96} x^3 \right. \\ & - \frac{1}{3840} x^5 \left. \right) \ln(x) + \frac{5 B \left(1 - \frac{1}{12} x^2 + \frac{1}{384} x^4 - \frac{1}{23040} x^6 \right)}{x} \\ & - \frac{20 (b_0 + b_1 x + b_2 x^2 + b_3 x^3 + b_4 x^4 + b_5 x^5 + b_6 x^6)}{x^5} \\ & + \frac{5 (b_1 + 2 b_2 x + 3 b_3 x^2 + 4 b_4 x^3 + 5 b_5 x^4 + 6 b_6 x^5)}{x^4} + x \left(B \left(1 - \frac{1}{12} x^2 \right. \right. \\ & \left. \left. + \frac{1}{384} x^4 - \frac{1}{23040} x^6 \right) \ln(x) \right. \\ & \left. + \frac{b_0 + b_1 x + b_2 x^2 + b_3 x^3 + b_4 x^4 + b_5 x^5 + b_6 x^6}{x^4} \right) \end{aligned} \quad (4.18)$$

> `eq5 := expand(eq4)`

$$\begin{aligned} eq5 := & \frac{4B}{x} + b_3 + 5b_5 - \frac{1}{23040} B \ln(x) x^7 - \frac{2}{3} x B + 12 x b_6 + \frac{1}{32} B x^3 \\ & - \frac{1}{1440} B x^5 - \frac{3 b_1}{x^4} - \frac{4 b_2}{x^3} - \frac{3 b_3}{x^2} + \frac{b_0}{x^3} + \frac{b_1}{x^2} + \frac{b_2}{x} + x b_4 + x^2 b_5 + x^3 b_6 \end{aligned} \quad (4.19)$$

O termo envolvendo logaritmo deve ser desprezado, pois envolve uma potência além de 6. Devemos

agora igualar a zero os coeficientes de x :

$$\begin{aligned} > \text{eq6} := \text{expand}\left(x^4 \cdot \left(\text{eq5} + \frac{1}{23040} B \ln(x) x^7\right)\right) \\ \text{eq6} := 4 B x^3 + x^4 b_3 + 5 b_5 x^4 - \frac{2}{3} B x^5 + 12 b_6 x^5 + \frac{1}{32} x^7 B - \frac{1}{1440} x^9 B - 3 b_1 \\ - 4 b_2 x - 3 b_3 x^2 + x b_0 + x^2 b_1 + x^3 b_2 + x^5 b_4 + x^6 b_5 + x^7 b_6 \end{aligned} \quad (4.20)$$

$b1 := 0$:

```
> eqs2 := array(1 .. 7);
for n from 1 to 7 do
    eqs2[n] := coeff(eq6, x, n) = 0
end do;
coeffEqs := convert(eqs2, list);
    eqs2 := array(1..7, [ ])
    eqs2_1 := -4 b_2 + b_0 = 0
    eqs2_2 := -3 b_3 + b_1 = 0
    eqs2_3 := 4 B + b_2 = 0
    eqs2_4 := b_3 + 5 b_5 = 0
    eqs2_5 := -\frac{2}{3} B + 12 b_6 + b_4 = 0
    eqs2_6 := b_5 = 0
    eqs2_7 := \frac{1}{32} B + b_6 = 0
```

$$\text{coeffEqs} := \left[-4 b_2 + b_0 = 0, -3 b_3 + b_1 = 0, 4 B + b_2 = 0, b_3 + 5 b_5 = 0, -\frac{2}{3} B + 12 b_6 + b_4 = 0, b_5 = 0, \frac{1}{32} B + b_6 = 0 \right] \quad (4.21)$$

```
> s2 := [B, seq(b[n], n = 1 .. 6)];
    s2 := [B, b_1, b_2, b_3, b_4, b_5, b_6] \quad (4.22)
```

$\text{sol2} := \text{solve}(\text{coeffEqs}, s2)$

$$\text{sol2} := \left[\left[B = -\frac{1}{16} b_0, b_1 = 0, b_2 = \frac{1}{4} b_0, b_3 = 0, b_4 = -\frac{25}{384} b_0, b_5 = 0, b_6 = \frac{1}{512} b_0 \right] \right] \quad (4.23)$$

$\text{assign}(\text{sol2})$

Como devemos obter somente uma solução particular, fazemos $b_0 = 1$, de modo que

$$b_0 := 1 \quad b_0 := 1 \quad (4.24)$$

$y2 := \text{expand}(y2)$

$$y2 := -\frac{1}{16} \ln(x) + \frac{1}{192} \ln(x) x^2 - \frac{1}{6144} \ln(x) x^4 + \frac{1}{368640} \ln(x) x^6 + \frac{1}{x^4} \quad (4.25)$$

$$+ \frac{1}{4x^2} - \frac{25}{384} + \frac{1}{512} x^2$$

A solução geral é então, até $O(6)$,

$$\begin{aligned} &> \mathbf{Y := C[1]*y1+C[2]*y2;} \\ Y := &C_1 \left(1 - \frac{1}{12} x^2 + \frac{1}{384} x^4 - \frac{1}{23040} x^6 \right) + C_2 \left(-\frac{1}{16} \ln(x) + \frac{1}{192} \ln(x) x^2 \right. \\ &\left. - \frac{1}{6144} \ln(x) x^4 + \frac{1}{368640} \ln(x) x^6 + \frac{1}{x^4} + \frac{1}{4x^2} - \frac{25}{384} + \frac{1}{512} x^2 \right) \end{aligned} \quad (4.26)$$

Vejam como o Maple resolve esta equação.

$$\begin{aligned} &> \mathbf{ics := y3(2) = 1, (D(y3))(2) = -1;} \\ &> \mathbf{eq4 := x*(diff(y3(x), x, x))+5*(diff(y3(x), x))+x*y3(x)=0;} \\ eq4 := &x \left(\frac{d^2}{dx^2} y_3(x) \right) + 5 \left(\frac{d}{dx} y_3(x) \right) + x y_3(x) = 0 \end{aligned} \quad (4.27)$$

$$\begin{aligned} &> \mathbf{dsolve(\{eq4, ics\})} \\ y_3(x) = &-\frac{4 (\text{BesselY}(2, 2) - \text{BesselY}(1, 2)) \text{BesselJ}(2, x)}{(-\text{BesselY}(2, 2) \text{BesselJ}(1, 2) + \text{BesselY}(1, 2) \text{BesselJ}(2, 2)) x^2} \\ &+ \frac{4 (\text{BesselJ}(2, 2) - \text{BesselJ}(1, 2)) \text{BesselY}(2, x)}{(-\text{BesselY}(2, 2) \text{BesselJ}(1, 2) + \text{BesselY}(1, 2) \text{BesselJ}(2, 2)) x^2} \end{aligned} \quad (4.28)$$

$$> P_{maple} := \text{plot}(op(2, \%), x = 0.8..3) :$$

Façamos uma comparação com o resultado obtido anteriormente:

$$\begin{aligned} &> \mathbf{e1 := subs(x=2, Y) = 1;} \\ e1 := &\frac{127}{180} C_1 + C_2 \left(-\frac{127}{2880} \ln(2) + \frac{13}{192} \right) = 1 \end{aligned} \quad (4.29)$$

$$\begin{aligned} &> \mathbf{e2 := subs(x=2, diff(Y, x)) = -1;} \\ e2 := &-\frac{31}{120} C_1 + C_2 \left(-\frac{581}{2880} + \frac{31}{1920} \ln(2) \right) = -1 \end{aligned} \quad (4.30)$$


$$\begin{aligned} &> \mathbf{s3 := fsolve(\{e1, e2\})} \\ s3 := &\{C_1 = 1.228744083, C_2 = 3.582227923\} \end{aligned} \quad (4.31)$$

$$> \mathbf{assign(s3) :}$$

$$> P_2 := \text{plot}(Y, x = 0.8..3, color = blue) :$$

$$> \mathbf{with(plots) :}$$

$$> \mathbf{display([P_{maple}, P_2])}$$



Vemos que há excelente concordância neste intervalo.